

〈資料〉 シュミット・ネットをマイコン用プロッターで
描くプログラム

BASIC Program to draw a Schmidt's net


山路 敦*
Atsushi YAMAJI

〈1〉 概要

1. プロッターを使ってシュミット・ネットを描くマイコン用プログラム。
2. 基円の半径 r は自由に指定できる ($r < 11.5 \text{ cm}$)。
3. A 3 版の用紙の任意の位置にネットを描ける。

作図例を第 1 図に掲げる。

(注) 半径 $r \leq 3 \text{ cm}$ の場合, シュミット・ネットのメッシュは 10 度。
 $r > 3 \text{ cm}$ では, 2 度。



第 1 図

< 2 > 使用機種

マイコン：NEC PC-9801

プロッター：渡辺測器製 マイプロットⅡ MP1000-31型(8BIT-D)

< 3 > 使用法

1. A3版の紙をプロッターにセットする。
2. 基円の半径 r を cm 単位で入力する ($r < 11.5$ cm)。
3. 基円の中心の x 座標を入力する ($0 < x < 36$ cm)。
4. 基円の中心の y 座標を入力する ($0 < y < 27$ cm)。
5. 入力の誤りがある場合には、Rキーを押すと半径の入力に戻る。
6. 経緯度で10度ごとの太線用のペンをペン・ストック1にセットする。
2度ごとの細線用のペンをペン・ストック2にセットする。
7. リターン・キーを押すと、作図が始まる。

< 4 > シュミット・ネットを描く方法

半径 R の半球面を考え、その上の点 P の位置を経度 ϕ 、余緯度 θ で表す。半球面の極 O が $\theta = 0$ となるように座標をとる(第2図)。極 O に接する平面の上にシュミット・ネットを描く。点 P に対応する点を \bar{P} とする。平面上の点の位置は、点 O を中心とする極座標 (Ψ, ρ) で表現する。シュミット・ネットは等積投影だから、半球面上に任意にとった2つの閉曲線領域の面積の比が投影に関して不変でなければならない。この投影は、座標系 (ϕ, θ) から座標系 (Ψ, ρ) への座標変換、

$$\begin{cases} \rho = 2R \cdot \sin(\theta/2) \\ \Psi = \phi \end{cases} \dots\dots\dots (i)$$

で表される。

シュミット・ネットを描くには、第2図の点 P を $a = \text{一定}$ または $b = \text{一定}$ の曲線に沿って動かして、それに対応する点 \bar{P} の軌跡をたどればよい。そこで第2図のようにとった2つの座標系 (a, b) から (ϕ, θ) への変換が必要になる。その変換は球面三角法によつて、

$$\begin{cases} \cos(\theta) = \cos(a) \cdot \sin(b) \\ \sin(\phi) \cdot \sin(\theta) = \sin(a) \cdot \sin(b) \\ \cos(\phi) \cdot \sin(\theta) = \cos(b) \end{cases} \dots\dots\dots (ii)$$


と表される。

< 5 > プログラムリスト

```

10 REM   file name : stereonet
20 REM
30 CONSOLE 0.25:0.1:SCREEN 3.0:WIDTH 80
40 REM
50 REM
60   PI=3.14159:HPI=PI/2
70   RD=PI/180
80 REM
90   GOSUB *IPT
100  GOSUB *PENST
110  PRINT :PRINT "サクス ショウケン ヲ ヘンコウ スルナラ R ヲ"
120  PRINT "サクス ヲ ハシメル ナラハ" return key " ヲ オスコト"
130  S$=INPUT$(1)
140  IF S$="R" OR S$="r" THEN 90
150  GOSUB *PRICIR
160 REM
170 REM
180 IF R0>300 THEN STP=2:STP1=2:STP2=178:GOTO 200
190     STP=10:STP1=10:STP2=170
200 REM
210 REM 10° コトノケイセン
220 FOR A=10 TO 170 STEP 10
230     IF A=90 THEN 320
240     IF A MOD 20 =0 THEN C1=STP2:C2=STP1:C3=-STP:GOTO 260
250         C1=STP1:C2=STP2:C3= STP
260     B=C1 :GOSUB *SPTR
270     LPRINT "M":X.Y
280     FOR B=C1 TO C2 STEP C3
290         GOSUB *SPTR
300         LPRINT "D":X.Y
310     NEXT B
320 NEXT A
330 LPRINT "M":X0.Y0+R0 :LPRINT "D":X0.Y0-R0
340 LPRINT "D":X0.Y0-R0
350 REM 10° コトノケイセン
360 FOR B=10 TO 170 STEP 10
370 IF B MOD 20 =0 THEN C1=88 :C2=-88:C3=-2:GOTO 390
380     C1=-88:C2= 88:C3= 2
390     X=X0-(C3/2)*R0*SIN(B*RD):Y=Y0+R0*COS(B*RD)
400     LPRINT "M":X.Y
410     FOR A=C1 TO C2 STEP C3
420     IF A<>0 THEN GOSUB *SPTR:LPRINT "D":X.Y:GOTO 440
430     A=.0001: GOSUB *SPTR:LPRINT "D":X.Y:A=0
440     NEXT A
450     X=X0+(C3/2)*R0*SIN(B*RD):Y=Y0+R0*COS(B*RD)
460     LPRINT "D":X.Y
470 NEXT B
480 REM
490 REM
500 LPRINT "J":2
510 REM
520 REM 2° コトノケイセン
530 FOR A=STP1 TO STP2 STEP STP
540 REM IF A MOD 10 =0 THEN 670
550 IF A MOD 2*STP =0 THEN C0=20 :C1=20+STP :C2=160:C3=STP :GOTO 570
560     C0=160:C1=160-STP:C2=20 :C3=-STP
570     B=C0 :GOSUB *SPTR
580     LPRINT "M":X.Y
590     FOR B=C1 TO C2 STEP C3
600     GOSUB *SPTR
610     LPRINT "D":X.Y

```



第 2 図

```

620 NEXT B
630 NEXT A
640 REM 2° コート ノ イセシ
650 FOR B= 2 TO 178 STEP 2
660 REM IF B MOD 10 =0 THEN 810
670 IF B MOD 4 =2 THEN C1=88 :C2=-88:C3=-2:GOTO 690
680 C1=-88:C2= 88:C3= 2
690 X=X0-(C3/2)*R0*SIN(B*RD):Y=Y0+R0*COS(B*RD)
700 LPRINT "M":X,Y
710 FOR A=C1 TO C2 STEP C3
720 IF A<>0 THEN GOSUB *SPTR:LPRINT "D":X,Y:GOTO 740
730 A=.0001: GOSUB *SPTR:LPRINT "D":X,Y:A=0
740 NEXT A
750 X=X0+(C3/2)*R0*SIN(B*RD):Y=Y0+R0*COS(B*RD)
760 LPRINT "D":X,Y
770 NEXT B
780 REM
790 LPRINT "H"
800 END
810 REM
820 REM
830 REM
840 REM
850 REM ----- サクス ショウケン
860 *IPT
870 CLS
880 PRINT "サクス ショウケン センチ・メーター タイ テー ニュウリョク スルコト"
890 PRINT
900 INPUT "ステレオ・ネット ノ ハンケイ ( 0<R<11.5 ) ":R0
910 IF R0=>11.5 THEN PRINT "ハンケイ オオキスル ノテ、モウイチト ニュウリョク セヨ":GOTO 890
920 PRINT
930 PRINT "チュウシン ノ x-サマヒョウ (":R0+2:"<x<":34-R0:") "":INPUT X0
940 IF X0-R0<=2 THEN PRINT "ヒタリ ニ ハミタタス ノテ、モウイチト ニュウリョク セヨ":GOTO 920
950 IF X0+R0>=34 THEN PRINT "ミキ ニ ハミタタス ノテ、モウイチト ニュウリョク セヨ":GOTO 920
960 PRINT
970 PRINT "チュウシン ノ y-サマヒョウ (":R0+2:"<y<":25-R0:") "":INPUT Y0
980 IF Y0+R0>=25 THEN PRINT "ウエ ニ ハミタタス ノテ、モウイチト ニュウリョク セヨ":GOTO 960
990 IF Y0-R0<=2 THEN PRINT "シタ ニ ハミタタス ノテ、モウイチト ニュウリョク セヨ":GOTO 960
1000 REM
1010 R0=R0*100
1020 X0=X0*100
1030 Y0=Y0*100
1040 RETURN
1050 REM ----- ヘン・ストック
1060 *PENST
1070 PRINT:PRINT
1080 PRINT "10° コート ノ セン ヲ カク ヘン ヲ、ヘン・キャリッジ 1 ニ セット スルコト"
1090 PRINT "2° コート ノ セン ヲ カク ヘン ヲ、ヘン・ストック 2 ニ セット スルコト"
1100 RETURN
1110 REM ----- Primitive circle
1120 *PRICIR
1130 LPRINT "W":X0,Y0,R0,R0,900,8100
1140 LPRINT "M":X0+R0,Y0:LPRINT "D":X0-R0,Y0
1150 LPRINT "M":X0,Y0+R0:LPRINT "D":X0,Y0-R0
1160 RETURN
1170 REM ----- キュウメン (a,b)カラ ネット(X,Y) ノ ヘンカン
1180 *SPTR
1190 REM a,b : phi,thet
1200 CA =COS(A*RD):CB =COS(B*RD)
1210 SA =SIN(A*RD):SB =SIN(B*RD)
1220 SASB =SA *SB
1230 PHI =ATN(SASB /CB )
1240 IF SASB <0 THEN PHI =PHI +PI
1250 THET =ATN(SA /(CA *SIN(PHI )))
1260 R=R0*SIN(THET /2) *1.41421
1270 Y= R*COS(PHI ) + Y0
1280 X= R*SIN(PHI ) + X0
1290 RETURN

```